if (isMultiplayer) then {
	If (isServer) then {
		[] spawn {
			TYR_WEATHER_SYNC = [];
			while {true} do {
				TYR_WEATHER_SYNC set [0, daytime];
				TYR_WEATHER_SYNC set [1, windStr];
				TYR_WEATHER_SYNC set [2, windDir];
				TYR_WEATHER_SYNC set [3, lightnings];
				TYR_WEATHER_SYNC set [4, rain];
				TYR_WEATHER_SYNC set [5, overcast];
				TYR_WEATHER_SYNC set [6, fog];
				publicVariable "TYR_WEATHER_SYNC";
				sleep 14.2345;
			};
		};
	};

	if (!isDedicated) then {
		[] spawn {
			private ["_value"];
			waituntil{ sleep 0.2345; !isnil "TYR_WEATHER_SYNC"};
			sleep 1;
			diag_log "WEAHTER JIP: Start";
			while {true} do {
				_value = TYR_WEATHER_SYNC select 4;
				0 setRain _value;
				while {abs(Rain - _value) > 0.2} do { sleep 0.345; }; 
				_value = TYR_WEATHER_SYNC select 6;
				0 setFog _value;
				while {abs(Fog - _value) > 0.2} do { sleep 0.345; }; 
				_value = TYR_WEATHER_SYNC select 3;
				0 setLightnings _value;
				while {abs(Lightnings - _value) > 0.2} do { sleep 0.345; }; 

				0 setWindStr (TYR_WEATHER_SYNC select 1);
				0 setWindDir (TYR_WEATHER_SYNC select 2);
				
				_value = TYR_WEATHER_SYNC select 5;
				0 setOvercast _value;
				while {abs(overcast - _value) > 0.2} do { sleep 3.345; }; 
				// diag_log Format["WEATHER JIP: %1",[overcast, Rain, Fog, Lightnings, WindStr]];
				sleep 30.2345;
			};
		};
	};
};